

One Bar Chart, Two Variables, Three Axes

Anne Horney and Gail F. Kirk
Cooperative Studies Program Coordinating Center
VA Medical Center, Perry Point, Maryland

ABSTRACT

It is often useful to show two variables with different ranges of values on the same bar chart. For a study of hypertension we needed a bar chart showing both systolic and diastolic blood pressure on different scales and grouped within antihypertensive treatment regimens. This was a problem since bar charts produced by GCHART have only one response axis. We will show how the Annotate facility can be used with GCHART to produce a vertical bar chart with a left and a right axis.

For our study the mean systolic blood pressure was to be plotted against the left vertical axis and the diastolic against the right axis. We also wanted each bar labeled with its mean and 95% confidence interval. Systolic values were plotted in the usual manner with the axis ranging from 130 to 154. We adjusted the diastolic values to fit within the same range on the left axis and printed the right vertical axis with values ranging from 80 to 92 by using the Annotate facility. We also used the Annotate data set to draw 95% confidence intervals and to label each bar with the mean blood pressure value.

INTRODUCTION

For a study of efficacy of antihypertensive treatment regimens we needed bar charts of mean systolic and diastolic blood pressure. Originally we produced two very simple charts showing the mean blood pressure for each treatment group. The systolic blood pressure chart is shown in Figure 1. Later we received requests to produce a combination chart showing both blood pressures, grouped within treatment regimens. Systolic and diastolic were to be graphed on different scales, and the bars were to be labeled with their means and 95% confidence intervals.




Figure 1


THE TWO VARIABLE BAR CHART

In SAS/GRAPH® software graphing two variables on the same bar chart with different scales was a problem since bar charts produced by the procedure GCHART have only one response axis. We used PROC GCHART to produce a vertical bar chart of systolic values in the usual manner. Next we adjusted (scaled) the diastolic values and placed them on the same chart. Then we labeled the right vertical axis with actual diastolic values, drew the 95% confidence intervals and labeled the bars using the annotate facility. See Figure 2.

Data

Input data for the bar chart are systolic blood pressure means and standard errors and diastolic blood pressure means and standard errors by treatment regimen. The means and standard errors were calculated by the MEANS procedure in the usual way. Systolic values are charted as input. Adjusted diastolic values are charted, and original (input) values serve as labels. Diastolic

SYSTOLIC AND DIASTOLIC BLOOD PRESSURE



The numbers at the top of the bars give the average values and the vertical lines in the bars indicate the 95% confidence intervals of the averages.

Figure 2

values are scaled to fit on the systolic axis by the formula:

$$ADBP = (DBP - DLO) * 2 + SLO$$

where

ADBP is adjusted diastolic value

DBP is original (input) diastolic value

DLO is lowest charted diastolic value, here 80 mmHg

SLO is lowest charted systolic value, here 130 mmHg

and the factor 2 is used because one unit on the diastolic axis is equivalent to 2 units on the systolic axis.

The 95% confidence intervals are calculated in the DATA step as LOW and HI:

$$\text{mean} \pm 1.96 * \text{standard error.}$$

Annotate data set

The annotate data set is used to label the bars, to draw the confidence intervals and to label the right axis. Mean blood pressure values are placed on the chart over each bar using the annotate option FUNCTION='LABEL'. For systolic bars the mean value (SYS) is both the TEXT and the Y-axis position variable. Diastolic bars are labeled with the input mean value (DIA) and positioned according to the adjusted or scaled value BP.

Confidence intervals are drawn using the annotate options FUNCTION='MOVE' and FUNCTION='DRAW'; also, FUNCTION='LABEL' draws tops and bottoms for the lines with the specification TEXT='-'.

The line that becomes the right vertical axis is produced by the procedure GCHART option FRAME. All other right axis characteristics are set up in the annotate data set. The axis labels are placed on the chart by the option FUNCTION='LABEL' at a constant X-axis position and at Y-axis positions corresponding to the appropriate systolic axis locations. The axis tick marks are drawn as a series of short lines with the options FUNCTION='MOVE' and FUNCTION='DRAW'. Another use of FUNCTION='LABEL' produces the axis label; note that it is rotated 90° in the negative direction to turn the text in toward the chart.

Response Axis (RAXIS) definition

Axis1 is the response (left vertical) axis; systolic blood pressure is plotted against Axis1. The ORDER= option is used to limit the value range from 130 to 154 mmHg. Systolic bars, as well as the axis label and the axis values are printed in blue. The response axis label is printed up the left side of the graph by setting the option ANGLE=90.

Bottom Axis

Axis2, which distinguishes between the systolic and the diastolic, is the midpoint axis (MAXIS). Since the two blood pressures are identified by color, this axis has no label and no values.

The treatment regimens are shown on Axis3, the group axis (GAXIS). The option ORDER= specifies the sequence of regimens. To make space for printing the right axis, the group axis must be shortened and moved to the left. Length and placement of the axis are controlled by the options LENGTH=, OFFSET=, and ORIGIN=. FOOTNOTE1 is the label for this axis.

Chart

The chart variable BP_G identifies each observation as systolic or diastolic. The height of each bar is BP which is either its mean systolic blood pressure or its mean diastolic blood pressure scaled to the 130-to-154 range. The variable BP is the SUMVAR. REGIMEN is the

GROUP variable. Thus the blood pressure bars, identified as systolic or diastolic by color (MIDPOINT patterning), are grouped within treatment regimens.

Furthermore, each bar is labeled with the appropriate blood pressure mean and 95% confidence interval. The range of systolic blood pressure means shown on the left axis is 130 mmHg to 154 mmHg, and the diastolic blood pressure means range from 80 mmHg to 92 mmHg on the right axis.

No LEGEND is needed because the colors blue and red distinguish between systolic and diastolic blood pressures.

CONCLUSION

Two variables with different ranges of values can be shown on the same bar chart: one chart displaying two variables on three axes.

ACKNOWLEDGMENTS

We would like to thank Dr. Joe Collins for his helpful comments, and Dr. Steve Bingham and Dr. H. Mitchell Perry, Jr, for their assistance during the development of this technique.

SAS and SAS/GRAPH are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

THE AUTHORS

Anne Horney
(410) 642-2411 ext. 5298

Gail F. Kirk
(410) 642-2411 ext. 5296

Mailing address for both:
CSPCC (151E)
VA Medical Center
P. O. Box 1010
Perry Point, MD 21902

APPENDIX

```

DATA SYSDIA;
  LENGTH REGIMEN $3;

  INPUT REGIMEN SYS S_STDERR DIA D_STDERR;

/* SYSTOLIC */
BP_G = 1;
BP   = SYS;
LOW  = SYS - 1.96 * S_STDERR;
HI   = SYS + 1.96 * S_STDERR;
OUTPUT;

/* DIASTOLIC */
BP_G = 2;
LOW  = DIA - 1.96 * D_STDERR;
HI   = DIA + 1.96 * D_STDERR;

/* ADJUST THE DIASTOLIC VALUES SO THEY CAN BE
PLOTTED USING THE LEFT VERTICAL AXIS */
BP   = (DIA-80) * 2 + 130;
LOW  = (LOW-80) * 2 + 130;
HI   = (HI -80) * 2 + 130;

  KEEP REGIMEN SYS DIA BP BP_G LOW HI;

  OUTPUT;
CARDS;
D   140.760   0.61684   81.8507   0.32517
B   141.184   1.05895   84.3010   0.57055
A   142.215   0.85550   85.9346   0.43341
C   148.955   0.69828   86.5319   0.37012
D+B 140.106   0.88744   83.3636   0.47170
D+C 146.274   0.87733   85.3087   0.47878
D+A 142.728   1.02340   85.0305   0.57762
;
RUN;

GOPTIONS DEVICE=WIN
          TARGET=WINPRTC
          ROTATE=LANDSCAPE
          FTEXT=SWISS
          FTITLE=SWISS
          HTEXT=.25 IN;

PATTERN1 VALUE=SOLID COLOR=BLUE;
PATTERN2 VALUE=SOLID COLOR=RED;

/* ANNOTATE DATA */

DATA ANNOSYS; SET SYSDIA END=EOF;
LENGTH FUNCTION COLOR STYLE $8 TEXT $100 ;
RETAIN COLOR 'BLACK' WHEN 'A' LINE 1;

  GROUP   = REGIMEN;
  MIDPOINT = BP_G;

  XC=REGIMEN; YSYS='2'; XSYS='2'; HSYS='5';
  POSITION='2';

/* MEAN BP TEXT */
FUNCTION='LABEL'; Y=BP; SIZE=2;
IF BP_G=1 THEN TEXT=PUT(SYS,5.1);
ELSE          TEXT=PUT(DIA,5.1);
OUTPUT;

/* LOWER CONFIDENCE INTERVAL */
/* LOWER LINE */
POSITION='5';
FUNCTION='MOVE'; Y=LOW;          OUTPUT;
FUNCTION='DRAW'; Y=BP+.2; SIZE=.07; OUTPUT;
/* LOWER BAR */
FUNCTION='LABEL'; Y=LOW+.1; SIZE=3;
TEXT='-';          OUTPUT;

/* UPPER CONFIDENCE INTERVAL */
/* UPPER LINE */
FUNCTION='MOVE'; Y=BP+1;          OUTPUT;
FUNCTION='DRAW'; Y=HI; SIZE=.07;  OUTPUT;
/* UPPER BAR */
FUNCTION='LABEL'; Y=HI+.1; SIZE=3;
TEXT='-';          OUTPUT;

IF EOF THEN DO;

/* RIGHT VERTICAL AXIS */

/* RIGHT AXIS VALUES */
FUNCTION='LABEL'; COLOR='RED';
HSYS='3'; XSYS='3'; YSYS='2'; SIZE=3;
POSITION='5';
X=91.8;
Y=154; TEXT='92'; OUTPUT;
Y=152; TEXT='91'; OUTPUT;
Y=150; TEXT='90'; OUTPUT;
Y=148; TEXT='89'; OUTPUT;
Y=146; TEXT='88'; OUTPUT;
Y=144; TEXT='87'; OUTPUT;
Y=142; TEXT='86'; OUTPUT;
Y=140; TEXT='85'; OUTPUT;
Y=138; TEXT='84'; OUTPUT;
Y=136; TEXT='83'; OUTPUT;
Y=134; TEXT='82'; OUTPUT;
Y=132; TEXT='81'; OUTPUT;
Y=130; TEXT='80'; OUTPUT;

/* RIGHT AXIS TICK MARKS */
COLOR='BLACK'; SIZE=.07; HSYS='5';

FUNCTION='MOVE'; X=88.9; Y=154; OUTPUT;
FUNCTION='DRAW'; X=89.5;          OUTPUT;
FUNCTION='MOVE'; X=88.9; Y=152; OUTPUT;
FUNCTION='DRAW'; X=89.5;          OUTPUT;
FUNCTION='MOVE'; X=88.9; Y=150; OUTPUT;
FUNCTION='DRAW'; X=89.5;          OUTPUT;
FUNCTION='MOVE'; X=88.9; Y=148; OUTPUT;
FUNCTION='DRAW'; X=89.5;          OUTPUT;
FUNCTION='MOVE'; X=88.9; Y=146; OUTPUT;
FUNCTION='DRAW'; X=89.5;          OUTPUT;
FUNCTION='MOVE'; X=88.9; Y=144; OUTPUT;
FUNCTION='DRAW'; X=89.5;          OUTPUT;
FUNCTION='MOVE'; X=88.9; Y=142; OUTPUT;
FUNCTION='DRAW'; X=89.5;          OUTPUT;
FUNCTION='MOVE'; X=88.9; Y=140; OUTPUT;
FUNCTION='DRAW'; X=89.5;          OUTPUT;
FUNCTION='MOVE'; X=88.9; Y=138; OUTPUT;
FUNCTION='DRAW'; X=89.5;          OUTPUT;
FUNCTION='MOVE'; X=88.9; Y=136; OUTPUT;
FUNCTION='DRAW'; X=89.5;          OUTPUT;
FUNCTION='MOVE'; X=88.9; Y=134; OUTPUT;
FUNCTION='DRAW'; X=89.5;          OUTPUT;
FUNCTION='MOVE'; X=88.9; Y=132; OUTPUT;

```

```

FUNCTION='DRAW'; X=89.5;          OUTPUT;
FUNCTION='MOVE'; X=88.9; Y=130;  OUTPUT;
FUNCTION='DRAW'; X=89.5;          OUTPUT;

/* RIGHT AXIS LABEL */
COLOR='RED'; HSYS='3'; XSYS='3'; YSYS='2';
FUNCTION='LABEL'; SIZE=1.9; POSITION='5';
ANGLE=-90;
X=97; Y=141.9;
TEXT=
'Mean Diastolic Blood Pressure (mm Hg)';
OUTPUT;

END;
RUN;

AXIS1
  LABEL=(C=BLUE ANGLE=90
'Mean Systolic Blood Pressure (mm Hg)')
  ORDER=(130 TO 154 BY 2)
  MAJOR=(WIDTH=1)
  MINOR=NONE
  VALUE=(C=BLUE);

AXIS2 LABEL= NONE
      VALUE= NONE;

AXIS3 ORDER=( 'D+B' 'D' 'B' 'A' 'D+A' 'D+C' 'C')
      LENGTH=77.0 PCT
      OFFSET=(2,2)
      ORIGIN=(11.9 PCT)
      LABEL=NONE;

FOOTNOTE1 'Antihypertensive Regimen';
FOOTNOTE2;
FOOTNOTE3 H=1.8
"The numbers at the top of the bars give the
average values and the vertical";
FOOTNOTE5 H=1.8
"lines in the bars indicate the 95% confidence
intervals of the averages.";

PROC GCHART DATA=SYSDIA ANNOTATE=ANNOSYS;
TITLE H=3
'SYSTOLIC AND DIASTOLIC BLOOD PRESSURE';
VBAR BP_G / SUMVAR=BP
DISCRETE
GROUP=REGIMEN
GSPACE=3
RAXIS=AXIS1
MAXIS=AXIS2
GAXIS=AXIS3
PATTERNID=MIDPOINT
NOLEGEND
FRAME;

RUN;
QUIT;

```